
第１章 花瓶の展開図出力のコンピュータによる自動化および花瓶にさす花づくり 
１．動機 

 私は大石先生の授業で行った壺の製作(展開図の作成からは手動)により積極的にコンピュータを用いることで展開図づくりを自動化することを

思いつきました。また、行列を用いて半円を回転させることにより(花瓶にさす)花を作成しました。 

 

２．花瓶の展開図出力のコンピュータによる自動化 

① 目標 

 作りたい花瓶の垂直な断面図(図 1-1)を製作、そこから 3 つの値、この断面図

の r1、r2 および、高さ h を測りプログラムに入力するだけで自動的に花瓶の展

開図が出力されるようにすること。 

 

 

 

② 展開図の解析 

 今回、私は花瓶を横向きに回転体の集合体としてモデル化しました。そのため 

花瓶の展開図は扇形となっています。扇形の解析は次のようになりました。 

 

③ 展開図出力を自動化するプログラムの製作 

 プログラムは十進 BASIC を用いて製作しました。その際、プログラムは次の 3 つの要素を含む必要があると考えました。 

・プログラム利用者に r1､r2､h を入力してもらう要素。 

・上の解析より r1 、r2、h から OA､OB､th を計算する要素。 

・中心角 th の円弧 2 つ、これら扇形を結ぶ線分 2 つを出力する要素。 

第三の要素では②までとは異なる解析を必要としたのでこれを以下に説明します。 

 

～中心角 th の円弧のプログラム～

 円弧はある円の円周の一部です。今回は円をある１点から等距離ｒ

に位置する点をたくさん打つことで出力しました。 

 
図 1-3 のように中心となるある１点を原点(0，0)とした場合、等距離

にある点 p の座標は(r＊cos th,＊sin th)となります。 

円を出力するにはこの角度thの変域を0° ≤ th ≤ 360°にし、点を打つ

間隔を細かくする必要があります。今回は0.1度ずつ点を打ちまし

た。 

 円弧は円周の一部を切り取った形なので角度thの変域を変えれば出

力されます。BASICでは、角度はラジアン表記なので度数表記に変換

する定数d(=3.14159/180)を挿入しました。また、２つの円弧は距離

ｒをそれぞれOA、OBに変換し、角度thはすでに計算によって定まっ

ているので、TO thという形にして 

FOR x1＝0 TO th STEP  0.1 

 PLOT POINTS:OA＊COS(x1＊d),OA＊SIN(x1＊d)  

 PLOT POINTS:OB＊COS(x1＊d),OB＊SIN(x1＊d) 

NEXT x1                    となります。    

 

～2 つの扇形を結ぶ線分 2 つ、のプログラム～ 

 2 つの扇形の端の座標は、図３より 

点 A；(x,y)=(OA＊COSth,OA＊SINth) 

点 B；(x,y)=(OB＊COSth,OB＊SINth) 

これを十進 BASIC で直線を出力するには、線分 1 本当たり両端の座

標を PLOT POINTS:x1,y1;x2,y2 と入力すれば出力されます。 

ゆえに 

PLOT LINES:OA＊COS(th＊d),OA＊SIN(th＊d) 

;COS(th＊d),SIN(th＊d) 

PLOT LINES:OA,0;OB,0         となります。 

 

～プログラムの利便化～ 

Ⅰ．展開図の平行移動 

 現在までのプログラムでは、円弧の中心となるある 1 点を原点(0，

図 1-2 よりth =
360°×𝑟1

𝑂𝐴
・・・① 

また⊿OAC と⊿OBD は相似なので OC:OD=r1:r2 となる。

OD=OC+h より、OC:(OC+h)=r1:r2 となる。これを変形し 

OC =
h×r1

𝑟2−𝑟1
・・・②となる。 

⊿OAC に三平方の定理を適用し𝑂𝐴2 = 𝐴𝐶2 + 𝑂𝐶2。②を代入し

AC=r1 より𝑂𝐴2 = 𝑟12 +
ℎ2+𝑟12

(𝑟2−𝑟1)2
。変形しOA =

𝑟1√(𝑟2−𝑟1)2+ℎ2

はじめに 

 私達は『本年度の大石明徳先生の「情報」の授業で行った“文様付き壺の製作”および同先生の(昨年度の)SSH 関連授業「プルーフⅡ」で

の活動を発展させれば、授業内容を遥かに超える“モノ作り”を実現できるのではないか』と思い今回の研究を始め、その成果をこのような論

文をまとめるに至りました。その結果(これはあくまで結果的にですが)、本論文の第１章、第２章の内容は“第３回貴大学数理工学コンテスト”

で優秀賞を受賞なさった私達の先輩の論文に(少しだけ)似たものとなってしまいました。しかし、第３章の内容は超オリジナルであり第１・２

章のそれも先輩たちの研究とは一味違った独創的なものとなりましたのでその点を御批判・御評価頂きたく貴コンテストに応募するに至りま

した。宜しくお願い申し上げます。 

図 1-3 

0)としています。BASIC において原点の座標は出力画面の左上となっ

ています。(図 1-4 矢印赤) 

 
 

また BASIC において 0°、角度が増加する方向は上のようになって

ます。(矢印黒の向き) 

そのため、角度 th の変域の幅が 90°以上になる、または変域の上限

が 90°以上、下限が 0°以下になるとき展開図が画面上から外れてし

います。そこで円弧の中心座標をｘ=a1、ｙ=b1 とし、プログラムの

最初にその値を定めるようにしました。そして前述した各プログラム

(赤文字)は各行(xy)という形で出力する図形の座標を定めているの

で、x の最後に+a1 を、y の最後に+b1 を追加するように改変し、展

開図の平行移動をできるようにしました。(下のプログラム 1-1 

510~700 参照) 

 

Ⅱ．糊代の追加 

 また、組み立てる際に展開図に糊代がある方がよいと考えました。

糊代は展開図の内側と横に付けるようにし、プログラムを組みました。 

 
図 1-5 のように展開図の内側の糊代は元の円弧(直径が小さい方)を一

定の割合 N1 を決めて縮小し、横側の糊代は元の円弧２つの角度を一

定の割合 N２を決めて増加させました。加えて、糊代部分を閉じるた

め線分を１本追加しました。(プログラム 1-1 450～660 行 参照) 

 

Ⅲ．その他 

・さらに、r1=r2 のとき②の式がゼロ除算となるため条件節を挿入し

ました。 

・また r1>r2 のとき th、OA､OB の値が負の値となってしまうので、

r3、r4 という変数を作り、 

r1>r2 のとき r3=r2、r4=r1  

r1<r2 のとき r3=r1、r4=r2 としました。 

そして th、OA、OB を求めるプログラムの r1 を r3 に、r2 を r4 に修

正しました。(プログラム 1-1 240~410 参照) 

・また、せっかくなので文様付きの花瓶を製作したいと思い、プログ

ラムの最初に、花瓶に付けたい文様の画像を画面に呼び出す１文、

gload”F:(画像名).png”をはじめに挿入しました。

 

＜プログラム 1-1＞ 

 
 

３．花瓶にさす花づくり 

①目標 

 行列による回転移動で花を作成する。 

 

② 行列による回転移動  

 図 1-6 よりある１点Ｐ(x,y)を角度 th 回転移動させる場合を 

考えます。まず点 R(x,0)、Q(0,y)をとります。 

次に角 th 回転させた R’、Q’の座標は 

R’(xCOSth,yCOSth)、Q’(-SINth,yCOSth)となります。(左の図) 

そして右図より P’の座標は 

x1=xCOSth-ySINth、y1=xSINth+yCOSth となります。 

よって回転移動の式は右図より、 

(
𝑥1
𝑦1

) = (
COSth −SINth
SINth COSth

) (
𝑥
𝑦) となります。 

０° 

９０° 

100 CLEAR 

110 gload"F:小橋口文様.png" 

120 SET BITMAP SIZE 1000,662 

130 SET WINDOW 0,1000,662,0 

140 LET d=3.14159/180 

150 SET POINT COLOR 1 

160 SET LINE COLOR 1 

170 SET POINT STYLE 1 

180 LET a1=250 

190 LET b1=250 

200 LET N1=1.05 

210 INPUT  PROMPT "上底r1の値は":r1 

220 INPUT  PROMPT "上底r2の値は":r2 

230 INPUT  PROMPT "幅hの値は":h 

240 IF r1=r2 THEN 

250    LET L=2*r1*d*180 

260    PRINT "Lの長さは";L 

270    PLOT LINES:a1,b1;a1+L,b1 

280    PLOT LINES:a1+L,b1;a1+L,b1+h 

290    PLOT LINES:a1,b1;a1,b1+h 

300    PLOT LINES:a1,b1+h;a1+L,b1+h 

310 ELSE 

320   IF r1>r2 THEN  

330      LET r4=r1 

340      LET r3=r2 

350   ELSE  

360      LET r3=r1 

370      LET r4=r2     

380   END IF 

390 LET OA=r3*SQR((r4-r3)^2+h^2)/(r4-r3) 

400 LET OB=r4*SQR((r4-r3)^2+h^2)/(r4-r3) 

410 LET th=360*(r4-r3)/SQR((r4-r3)^2+h^2) 

420 PRINT "半径OAの長さは";OA 

430 PRINT "半径OBの長さは";OB 

440 PRINT "中心角θの大きさは";th 

450 IF r1>r2 THEN  

460       LET OC=OA*0.8 

470   ELSE  

480       LET OC=OB*1.1 

490   END IF 

500   FOR x1=0  TO th  STEP 0.1 

510 PLOT 

POINTS:OA*COS(x1*d*N1)+a1,OA*SIN(x1*d*N1)+b1 

520 PLOT 

POINTS:OB*COS(x1*d*N1)+a1,OB*SIN(x1*d*N1)+b1 

530 PLOT POINTS:OC*COS(x1*d)+a1,OC*SIN(x1*d)+b1 

540   NEXT x1 

550 IF h>0 THEN    

560   LET x1=OC*COS(th*d)+a1 

570   LET y1=OC*SIN(th*d)+b1 

580   LET x2=OB*COS(th*d)+a1 

590   LET y2=OB*SIN(th*d)+b1 

600   PLOT LINES:x1,y1;x2,y2 

610   PLOT 

LINES:OC+a1,b1;OB+a1,b1 

620   LET x1=OA*COS(th*d*N1)+a1 

630   LET y1=OA*SIN(th*d*N1)+b1 

640   LET x2=OB*COS(th*d*N1)+a1 

650   LET y2=OB*SIN(th*d*N1)+b1 

660   PLOT LINES:x1,y1;x2,y2  

670   LET x1=OA*COS(th*d)+a1 

680   LET y1=OA*SIN(th*d)+b1 

690   LET x2=OB*COS(th*d)+a1 

700   LET y2=OB*SIN(th*d)+b1 

710   PLOT LINES:x1,y1;x2,y2 

720   PLOT 

LINES:OA+a1,b1;OB+a1,b1 

730  END IF         

740 END IF 

750 END 

＜プログラム 1-2＞ 

 

 

③ 花のプログラム 

今回は半径 20 の半円を回転させることで花を作りました。半円を描くプログラムは右のようになります。

（プログラム 1-2）この半円を②の式でそのまま回転させると回転させた半円同士が重なり、結果として円が

出力されてしまいます。そこで、回転の中心と半円の中心を異なる点にとりました。(その距離は７５)を移動

させるようにしました。そのプログラムは次のようになります。 

FOR th0=0 TO 360 STEP 30 

 FOR th1=0 TO 180 STEP 0.1 

    LET x=20＊COS(th1＊d) 

LET y=20＊SIN(th1＊d) 

    PLOT POINTS:x＊COS(th0＊d)-y＊SIN(th0＊d)+200+75＊COS(th0＊d),x＊SIN(th0＊d)+y＊COS(th0＊d)+250+75＊SIN(th0＊d) 

 NEXT th1 

NEXT th0 

 

しかし、このプログラムの実行結果は図 1-7A となり閉じた形にはなりません。これは半円の th1

の変域が0° ≤ th1 ≤ 180°となっているためです。（図 1-7A の緑丸の中) 

そこで、変域を−90° ≤ th1 ≤ 90°にします。すると図 1-7B が出力され閉じた形(花の形)なりま

す。(図 1-7B の緑丸の中) 以上を基に作成したプログラム(プログラム 1-3)とその実行結果(図 1-

7C)をここに示しておきます。 

 

＜プログラム 1-3＞ 

 
 

 

第 2 章 画像の湾曲化による花瓶の表面への画像の貼り付け 

1   研究の動機と目的 

 第１章では思い通りの形と模様の花瓶の展開図を作り出す方法を考えました。しか

しこの方法だと円錐台の境目で模様が切れてしまい、模様の連続性がなくなってしま

います。僕は画像を湾曲化させれば模様が切れることがなくなるのではないかと考え

ました。そして湾曲化させることで、図 2-1 の右の花瓶のように花瓶の曲面に絵を貼り

付けることができるのではないかと考えたのです。 

 この章では画像を変形させ花瓶の曲面上でつなぐことを目的とします。 

 また展開図によって x 座標、y 座標の範囲がばらつくので、展開図の x 座標、y 座標

の最大値、最小値をあらかじめ計算しておき、展開図が出力後の画像に入るように画像

のトリミングをさせることにしました。 

 

FOR th=0 TO 180 
STEP 0.1 

 LET x=20＊COS(th＊d) 

LET y=20＊SIN(th＊d) 

  NEXT th               

図 1-7A 図 1-7B 

100 CLEAR 

110 SET BITMAP SIZE 1016,662 

120 SET WINDOW 0,1000,652,0 

130 LET d=3.14159/180 

140 SET POINT STYLE 1 

150 SET LINE COLOR 1 

160 SET POINT COLOR 24 

170 FOR th0=0 TO 360 STEP 30 

180   FOR th1=-90  TO 90  STEP 0.1 

190      LET x=20*COS(th1*d) 

200      LET y=20*SIN(th1*d) 

210      LET x1=x*COS(th0*d)-y*SIN(th0*d)+500+75*COS(th0*d) 

220      LET y1=x*SIN(th0*d)+y*COS(th0*d)+250+75*SIN(th0*d)     

230     PLOT POINTS:x1,y1 

240    NEXT th1 

250 NEXT th0       

260 SET AREA COLOR 27 

270 flood 500,250 

280 FOR th0=0 TO 360 STEP 30 

290    FOR th1=-90  TO 90  STEP 0.1 

300       LET x=20*COS(th1*d) 

310       LET y=20*SIN(th1*d) 

320       LET x1=x*COS(th0*d)-y*SIN(th0*d)+500+50*COS((th0+10)*d) 

330       LET y1=x*SIN(th0*d)+y*COS(th0*d)+250+50*SIN((th0+10)*d) 

340      PLOT POINTS:x1,y1 

350   NEXT th1 

360 NEXT th0       

370 SET AREA COLOR 6 

380 flood 500,250 

390 FOR th0=0 TO 360 STEP 0.1 

400  PLOT POINTS:50*COS(th0*d)+500,50*SIN(th0*d)+250 

410 NEXT th0       

420 SET AREA COLOR 22 

430 flood 500,250 

 

440 FOR th0=0 TO 360 STEP 30 

450   FOR th1=-90  TO 90  STEP 0.1 

460   LET x=20*COS(th1*d) 

470    LET y=20*SIN(th1*d) 

480    LET x1=x*COS(th0*d)-y*SIN(th0*d)+500+25*COS((th0+20)*d) 

490    LET y1=x*SIN(th0*d)+y*COS(th0*d)+250+25*SIN((th0+20)*d)  

500      PLOT POINTS:x1,y1 

510   NEXT th1 

520 NEXT th0       

 

530 END 

図 1-7C 

2   方法 

②画像のトリミング 

OA、OB、th の値によっては展開図が描かれる座標が大きくなり、ウィンドウの座標の範囲を固定していると表示できる領域からはみ出すときが

あります。そのため図 2-4、表 2-1 のように x1、y1 の最大値、最小値を求めてウィンドウ上に表示される座標をしぼります。 

 

④ プログラムの内容 

① と②を使ってプログラムを作成しました。プログラムは次のページに記載しています(プログラム 2-1)。 

まずは円錐台の個数、半径、高さをそれぞれ変数 Hm、配列 haneki(Hm+1)、配列 hs に代入します(1000~1240 行)。なお「DIM 文」は配列作る

文です。初めにすべての花瓶の寸法を打ち込むのは、本プログラムでは画像を分割する必要があるからです。そして、第１章でも触れた「gload

文」を使い window 上に画像を表示し(1340 行)、二次元配列 xy(500,500)に縦横 500 に区切った色指標を「ASK PIXEL VALUE 文」で代入しま

す(1360~1420 行)。この時点で画像は 500*500[px]で扱われます。そして上の円錐台から順番に展開図の出力を開始します(1430~2610 行名)。出

力する円錐台の上底と下底に応じて配列 haneki から r1(上底)、r2(下底)を取り出し OA、OB、th を計算します。その際 r2 よりも r1 が大きいと

OA、OB、th が負の値になってしまうので、r1 と r2 の値を入れ替え、画像を上下反転させるかどうかを表す変数 hanten に 1 を代入します。ま

た r1 と r2 の値が等しいときは、展開図が長方形になるということを表す変数 rec に 1 を代入します (1450~1630 行)。r1 と r2 の値がほとんど

変わらないときなどは、展開図の座標が及ぶ範囲が広くなってしまうことがあります。そのため②で示したように、ウィンドウで表示する座標の

範囲を、展開図がどこまで及ぶかの値に合わせて指定しています(1650~1940 行)。1960~2540 行で点(小円)をプロットしています。まずは配列

xy に入れ込んだ色情報 BASIC の色指標に変換します(1980~2360 行)。そして rec=1 ならば円筒の局面の展開図を、そうでなければ上で求めた

数式を使って湾曲させたものを出力します。また hanten=1 ならば FOR 文で y が減少するように式を変形します(2370~2500 行)。上の色指標で

プロットする点は、半径１[px]の小円を用います(2510-2520 行)。x と y がそれぞれ 500 に達したら円錐台はウィンドウ上に出力されているので、

画像ファイルとして出力します(2560~2600 行)。この作業を円錐台の個数分だけ行います(1960~2610 行)。すべての作業が終わったらプログラ

ム終了(2620 行)。 

3   結果と考察 

 [画像 2-1]から[画像 2-2]のように展開図(bmp ファイル)が出力されました。その展開図をもとに作った花瓶が[画像 2-3]です。 

4   結論・今後の課題 

 自作のプログラムを使い展開図を作り、花瓶に画像を貼り付けることができました。しかし、AB が長くなると[画像 2-4]

のように小円と小円の間の隙間が大きくなってしまいます。そのためプロットする小円の半径を中心からの距離に応じて大

きくしようと考えました。しかし、弧方向では小円の中心からの距離によって小円の疎密が変わり、半径方向では AB の長

さによって変わります。小円の半径を変えると弧方向・半径方向の両方の疎密が同時に変わってしまうので具合が悪いので

す。現在は変形前の画像の 1 区切りごとに 1 つの点のみプロットしていますが、複数プロットすることによってこの問題を

解決できると考えています。 

 x1 の最大値 y1 の最大値 x1 の最小値 y1 の最小値 

①  OB OBsinθ OAcosθ 0 

②  OB OB OBcosθ 0 

③  OB OB -OB OBsinθ 

④  OB OB -OB -OB 

①画像を湾曲化させる数式 

第 1 章で、上底 r1、下底 r2、高さ h を使って

円錐台の展開図中の OB と OA そして中心角

th を求める計算式を導きました。 

図 2-2に従って元の画像の直交座標系の 1点を

展開図の一点に変換する数式を導きました。 

ただし、プログラム上では y 座標を反対方向に

とっているので若干異なる式になります。 

 

 

 

R1<R2 と R1>R2 の違い 

下の図 2-3 で表した通り R1<R2 のときは

画像の下側の半径が大きくR1>R2のときは

画像の上側の半径が大きい。 

上 

 

下 

下 

 

上 R1 < R2 のとき 

とき 

R2 < R1 のとき 

とき 

①θ <
π

2
 ②

π

2
≤ θ < π ③π ≤ θ <

3

2
π ④

3

2
π ≤ θ < 2π 

l 

l = OA +
𝐻𝑚 × y(𝑂𝐵 − 𝑂𝐴)

500
 

𝜃1 = 𝑡ℎ ×
500 − 𝑥

500
 

𝑥1 = l × cos𝜃1 

𝑦1 = l × 𝑠𝑖𝑛𝜃1 

l = OA −
𝐻𝑚 × 𝑦(𝑂𝐵 − 𝑂𝐴)

500
 

𝜃1 = 𝑡ℎ ×
𝑥

500
 

𝑥1 = l × cos𝜃1 

𝑦1 = l × 𝑠𝑖𝑛𝜃1 

𝑥1 = 2π𝑟1 ×
𝑥

500
 

𝑦1 = ℎ ×
𝑦

500
 

(x,y) 

(x１,y１) 

(x１,y１) 

(x１,y１) 

l 

l 

下 上 

上 下 

上 

 

下 

上 

下 

th 

th 

画像 2-1 
画像 2-2 

画像2-4 

画像 2-3 

図 2-2 

図 2-3 

図 2-4 
表 2-1 

つなぎ絵花瓶ちゃん.BAS ＜プログラム 2-1＞

1000 !'******* 花 瓶 の 情 報 入 力

************************************** 

1010 LET Hm =  10           !'**Hm:花瓶

を何個の円錐台に分けるか？ 

1020 DIM hankei(Hm+1)       !'**半径の配

列(Hm+1分) 

1030 LET hankei(1)=70       !'**半径入力

(Hm+1個分入力) 

1040 LET hankei(2)=44.5 

1050 LET hankei(3)=30 

1060 LET hankei(4)=24 

1070 LET hankei(5)=30 

1080 LET hankei(6)=44.5 

1090 LET hankei(7)=69.5 

1100 LET hankei(8)=85 

1110 LET hankei(9)=85 

1120 LET hankei(10)=69.5 

1130 LET hankei(11)=29.5 

1140 DIM hs(Hm)              !'**円錐台

の高さ 

1150 LET hs(1)=21.5 

1160 LET hs(2)=29.5 

1170 LET hs(3)=29.5 

1180 LET hs(4)=21.5 

1190 LET hs(5)=21.5 

1200 LET hs(6)=21.5 

1210 LET hs(7)=21.5 

1220 LET hs(8)=29.5 

1230 LET hs(9)=29.5 

1240 LET hs(10)=21.5   

1250 LET bairitu=2.2 

1260 FOR i=1 TO Hm+1 STEP 1         !'**

倍率設定 

1270   LET kari_han=hankei(i) 

1280   LET hankei(i)=kari_han*bairitu 

1290 NEXT i     

1300 FOR i=1 TO Hm STEP 1 

1310   LET kari_hs=hs(i) 

1320    LET hs(i)=kari_hs*bairitu 

1330 NEXT i         

1340 gload" 昼 の 学 校 （ 半 径 文

字）.png"        !'**貼り付ける画像 

1350 !'********************************* 

1360 DIM xy(0 TO 500, 0 TO 500) 

1370 SET COLOR MODE "NATIVE" 

1380 FOR y=0 TO 1 STEP 1/500 

1390    FOR x=0 TO 1 STEP 1/500 

1400       ASK PIXEL VALUE(x,1-y) 

xy(x*500,y*500) 

1410    NEXT x 

1420 NEXT y 

1430 FOR Wn=1 TO Hm STEP 1 

1440    !'******** r1,r2.h を取り出し

OA,OB,th,hanten,rec を 求 め る 

************* 

1450 LET r1 = hankei(Wn) 

1460 LET r2 = hankei(Wn+1) 

1470 LET h = hs(Wn) 

1480 LET hanten = 0 

1490 LET rec = 0 

1500 IF r1>r2 THEN 

1510   LET r3 = r1 

1520   LET r1 = r2 

1530   LET r2 = r3 

1540   LET hanten = 1 

1550 ELSEIF r1=r2 THEN 

1560   LET rec = 1 

1570   LET disksize = 1 

1580 END IF  

1590 IF rec = 0  THEN 

1600   LET OA=r1*SQR((r2-r1)^2+h^2)/(r2-

r1) 

1610   LET OB=r2*SQR((r2-r1)^2+h^2)/(r2-

r1) 

1620       LET th=2*PI*(r2-r1)/SQR((r2-

r1)^2+h^2) 

1630 END IF  

1640    !'****************************** 

1650    !'**** 画像のトリミング********* 

1660    IF rec = 0 THEN 

1670       LET rightmax = OB 

1680       IF th<(PI/2) THEN 

1690          LET topmax = SIN(th)*OB 

1700          LET leftmax = COS(th)*OA 

1710          LET bottommax = 0 

1720       ELSEIF th<PI THEN 

1730          LET topmax = OB 

1740          LET leftmax = COS(th)*OB 

1750          LET bottommax = 0 

1760       ELSEIF th<(3*PI/2) THEN 

1770          LET topmax = OB 

1780          LET leftmax = -OB 

1790          LET bottommax = SIN(th)*OB 

1800       ELSE 

1810          LET topmax = OB 

1820          LET leftmax = -OB 

1830          LET bottommax = -OB 

1840       END IF  

1850    ELSE 

1860       LET rightmax = r1*2*PI 

1870       LET topmax = h 

1880       LET leftmax = 0 

1890       LET bottommax = 0 

1900    END IF 

1910    !'****************************** 

1920    CLEAR 

1930    SET bitmap SIZE rightmax-

leftmax+4,topmax-bottommax+4 

1940    SET WINDOW leftmax-

2,rightmax+2,bottommax-2,topmax+2 

1950!'************FOR 文 x,y 

************************************ 

1960    FOR y=500*(Wn-1)/Hm TO 500*Wn/Hm 

STEP 1 

1970       FOR x=0 TO 500 STEP 1 

1980          DIM c$(0 TO 5) 

1990          DIM cc(0 TO 5) 

2000          !'---------------------      

2010          !'------------------------ 

2020          LET c = xy(x,y) 

2030          LET c1=c+16^6  

2040          LET c1$=BSTR$(c1,16) 

2050          LET b5$=mid$(c1$,2,1) 

2060          LET b4$=mid$(c1$,3,1) 

2070          LET g3$=mid$(c1$,4,1) 

2080          LET g2$=mid$(c1$,5,1) 

2090          LET r1$=mid$(c1$,6,1) 

2100          LET r0$=mid$(c1$,7,1) 

2110          LET c$(5)=b5$ 

2120          LET c$(4)=b4$ 

2130          LET c$(3)=g3$ 

2140          LET c$(2)=g2$ 

2150          LET c$(1)=r1$ 

2160          LET c$(0)=r0$ 

2170          FOR m=5 TO 0 STEP -1 

2180             IF c$(m)="A" THEN LET 

c$(m)="10" 

2190             IF c$(m)="B" THEN LET 

c$(m)="11" 

2200             IF c$(m)="C" THEN LET 

c$(m)="12" 

2210             IF c$(m)="D" THEN LET 

c$(m)="13" 

2220             IF c$(m)="E" THEN LET 

c$(m)="14" 

2230             IF c$(m)="F" THEN LET 

c$(m)="15" 

2240          NEXT m 

2250          FOR n=5 TO 0 STEP -1 

2260             LET cc(n)=VAL(c$(n)) 

2270          NEXT n  

2280          LET 

b=(16^1*cc(5)+16^0*cc(4))/256 

2290          LET 

g=(16^1*cc(3)+16^0*cc(2))/256 

2300          LET 

r=(16^1*cc(1)+16^0*cc(0))/256 

2310          !'------------------------ 

2320          SET COLOR mode "REGULAR" 

2330          !'------------------------ 

2340          SET COLOR MIX(255) r,g,b 

2350          SET AREA COLOR 255 

2360          !------------------------- 

2370          IF rec =1 THEN  

2380             LET x1=x*r1*2*PI/500 

2390            LET y1=(500*Wn/Hm-

y)*h/(500/Hm) 

2410          ELSE 

2420             IF hanten = 1 THEN  

2430                LET x1=COS(th*(500-

x)/500)*(OA+((OB-OA)*(-

y+500*Wn/Hm))/(500/Hm)) 

2440                LET y1=SIN(th*(500-

x)/500)*(OA+((OB-OA)*(-

y+500*Wn/Hm))/(500/Hm))       

2450             ELSE  

2460                LET 

x1=COS(th*x/500)*(OA+((OB-OA)*(y-

500*(Wn-1)/Hm))/(500/Hm)) 

2470                LET 

y1=SIN(th*x/500)*(OA+((OB-OA)*(y-

500*(Wn-1)/Hm))/(500/Hm))              

2480             END IF 

2490          END IF  

2500          !------------------------- 

2510          LET disksize =1 

2520          DRAW disk WITH 

SCALE(disksize)*SHIFT(x1,y1) 

2530       NEXT x 

2540    NEXT y 

2550    !'****************************** 

2560    DECLARE STRING filename$ 

2570!'******** 展 開 図 の 保 存 先

************************************* 

2580    LET  filename$ ="展開図出力先

/"&STR$(Wn)&".bmp"   

2590    !'****************************** 

2600    GSAVE filename$ 

2610 NEXT Wn 

2620 END 

第 3 章 縦切りの花瓶 

1. 研究の動機と目的 

 第 1 章、第 2 章では花瓶（回転体）を横に切って開いた展開図を

作りましたが、私はさらに、図 3-1 のように縦に切った場合の展開

図はどのように作成できるのか考えることにしました。横切りのと

きは、各段の展開図をそれぞれ計算して作る手間がありましたが、

縦切りなら全てのパーツが同じ形です。それが縦切りの良いところ

だと思います。 

 この研究では、花瓶を縦に切った場合の展開図をコンピュータで

出力する方法を考えます。 

 

2. 方法 

 十進 BASIC を使用し、展開図のプログラムを作成します。 

 まず、この花瓶を関数で表現します。 

 
 以降、正弦曲線の部分を「上部」、円弧の部分を「下部」と呼びます。 

 次に図 3-1 のように花瓶を展開可能な立体にモデル化し分割します。元の滑らかな花瓶では水平な断面は円になりますが、その

円に内接する正多角形を断面図とする図形にモデル化します。縦に分割された各パーツはその正多角形の一辺の集合体となります。 

 また、以下では展開図の枚数を n 枚として計算します。 

 各パーツをコンピュータで出力するには各パーツの幅と長さを求める必要があります。以下にその計算を示します。 

 

 
 

 まず、図 3-3、3-4 のように変数 h の関数 y,L を設定しま

す。後の出力の都合で y は幅全体の半分の長さになっていま

す。展開図の縦の中央線から半分ずつ出力することで、左右

対称な形の展開図にします。展開図の左右の曲線を別々のプ

ログラムで出力するということでもあります。 

 y,L を h の式で表すことができれば、それを用いて展開図を

描けるので、設定した a,b,c,n と h を用いて、幅 y と上端から

の長さ L を計算していきます。 

 

① 幅 y 

 図 3-2 にある通り、h,a,b,c を用いて元の滑らかな花瓶の水

平な断面図の半径(図 3-4 にある通り、この半径を r としてい

る)は次のように表せます。 

上部 r=a×cos( 
2π

c
×h)+a+b 

(正弦曲線の周期は c(c は半球を除いた高さ)なのでこの式にな

ります。) 

下部 r=√(2 × a + b)2－h2 (円の式である。) 

なので、次に展開図の幅 y は図 3-4 より、 

y=r×sin(π/n) 

上部 y= (a×cos(
2π

c
×h)+a+b)×sin(π/n) 

下部 y=√(2 × a + b)2－h2×sin(π/n) 

と表せます。 

 

② 長さ L 

 まず、長さを考える曲線は、ℓ=f(h)(ℓは図 3-4 参照)で表され

る曲線です。f(h)は具体的には、 

f(h)= r×cos(π/n) 

上部 f(h)=(a×cos(
2π

c
×h)+a+b)×cos(π/n) 

下部 f(h)=√(2 × a + b)2－h2×cos(π/n) 

と表せます。 

 そして、この曲線の長さを微積分を用いて求めることを試

みました。 

曲線 y=f(x) (a≦x≦b)の長さ L は、 

L=∫ √1 + {f ′(x)}2𝑏

a
𝑑𝑥 

です。これを用いて上部を計算してみます。, 

まず微分の部分は、 
dℓ

dh
  

=((a×cos(
2π

c
×h)+a+b)×cos(π/n))’ 

 

 

図 3-1 

図 3-2 

図 3-2 のように h,r,a,b,c を設定すると 

(h は変数であることに注意してください) 

(r は図 3-4 の通り断面の半径) 

(a は正弦曲線の振幅) 

(b は花瓶の最も細くなる部分の半径) 

(c は花瓶の正弦曲線の部分のみの高さ) 

正弦曲線は 

r= a×cos( 
2π

c
×h)+a+b (0≦h≦c) 

円弧は 

r=√(2 × a + b)2－h2 (0≦h≦2a+b) 

とそれぞれ表せる。 

ただし内接す

るようにモデ

ル化する。 

上

か

ら

見

た

図 

← 

回転 

y 

L 

h r 
π

n
rad 

ℓ y 

断面図 

図 3-3 

図 3-4 

2a+b 

2a+b 

r 

h 

正弦曲線 

円弧 

=cos(π/n)(a×cos(
2π

c
×h)+a+b)’ 

=cos(π/n)× 
2π

c
×(－a sin(

2π

c
×h)) 

となります。 

そして、 

L=∫ √1 + ( cos(π/n) ×
2π

c
× (－a sin( 

2π

c
× h)))

2
ℎ

0
𝑑ℎ 

 =∫ √1 + cos2(π/n) ×
4π

2

c2
× 𝑎2 × sin2( 

2π

c
× h)

ℎ

0
𝑑ℎ 

となるのですが、高校数学の範囲ではこの積分を計算するこ

とができませんでした。 

 なので次に、曲線を区切って直線の集まりと見て長さを求

める方法を考えました。 

 

③ 区切って長さを求める 

 このプログラムでは、既に求めた①幅の式も使用します。 

また、今まで L とおいていた展開図の長さの部分を、ここか

らは sum とおきます。 

 まず上部について説明します。はじめは sum=0 です。そし

て、sum,y(h=0 のときの y)に点を打ちます。次に、図 3-5 の

sum1 の長さを求め、sum に加算します。h の間隔は

c/100000 としています。そうしてまた

sum,y(h=
1

100000
𝑐のときの y)に点を打ちます。 

以下、それを繰り返して展開図を描きます。 

 これをプログラムにするためには、 

FOR NEXT 文の中に 

LET sum=sum+SQR((𝑐/100000)^2 + (f(h + 𝑐/100000)

－f(h))^2 ) 

(f(h)=ℓである。) 

という代入の文を入れたプログラムにします。 

(ℓは前に示したように変数 h の式です。そして SQR(~)の部分

が三平方の定理を利用して sum1,sum2,…を求めています。) 

 ただし、FOR NEXT 文の h の範囲を単に 0~c(半球を除い

た高さ)にすると最後の繰り返しで sum に関係する部分が余分

に計算されてしまい、sum にズレが生じます。そこで、h の

範囲を 0~(c－c/100000)とし、最後の h=c の時の点は改めて別

のプログラムで打つようにすることでこの問題を解決しまし

た。(プログラム 3-1 の 1120 行,1190 行,1200 行など) 

 また、花瓶の下部についても同様にプログラムを作成しま

す。下部については上記のような sum に関する余分な計算が

行われると、円の式で負の数の平方根を計算してしまうこと

が分かったので、上部と同じようにプログラムを修正しまし

た。

 
 出来上がったプログラムを以下に示します。

 

 
 

1290 LET y=SQR((2*a+b)^2-(h)^2)*SIN(d/n) 

1300 PLOT POINTS:sum+100,-y+200 

 

1310 LET sum=0 

1320 LET st=c/100000 

1330 FOR h=0 TO c-st STEP st 

1340    LET y=(a*COS(h/c*(2*d))+a+b)*SIN(d/n) 

1350    PLOT POINTS:sum+100,y+200 

1360    LET l2=(a*COS((h+st)/c*(2*d))+a+b)*COS(d/n) 

1370    LET l1=(a*COS(h/c*(2*d))+a+b)*COS(d/n) 

1380    LET sum=sum+SQR((st)^2+(l2-l1)^2) 

1390 NEXT h 

1400 LET y=(a*COS(h/c*(2*d))+a+b)*SIN(d/n) 

1410 PLOT POINTS:sum+100,y+200 

 

1420 LET st=(2*a+b)/100000 

1430 FOR h=0 TO 2*a+b-st STEP st 

1440    LET y=SQR((2*a+b)^2-(h)^2)*SIN(d/n) 

1450    PLOT POINTS:sum+100,y+200 

1460    LET l2=SQR((2*a+b)^2-(h+st)^2)*COS(d/n) 

1470    LET l1=SQR((2*a+b)^2-(h)^2)*COS(d/n) 

1480    LET sum=sum+SQR((st)^2+(l2-l1)^2) 

1490 NEXT h 

  

1500 LET y=SQR((2*a+b)^2-(h)^2)*SIN(d/n) 

1510 PLOT POINTS:sum+100,y+200 

 

1520 LET y=(2*a+b)*SIN(d/n) 

1530 PLOT LINES:100,y+200;100,-y+200 

  

1540 END 

1000 CLEAR 

1010 SET BITMAP SIZE 1016,662 

1020 SET WINDOW 0,1000,652,0 

1030 LET d=3.14159 

1040 SET POINT COLOR 1 

1050 SET POINT STYLE 1 

1060 INPUT  PROMPT "振幅 a の値は":a 

1070 INPUT  PROMPT "最も細いところの半径 b の値は":b 

1080 INPUT  PROMPT "底部の半球を除いた高さ c の値は":c 

1090 INPUT  PROMPT "展開図の枚数 n の値は":n 

 

1100 LET st=c/100000 

1110 LET sum=0 

1120 FOR h=0 TO c-st STEP st 

1130    LET y=(a*COS(h/c*(2*d))+a+b)*SIN(d/n) 

1140    PLOT POINTS:sum+100,-y+200 

1150    LET l2=(a*COS((h+st)/c*(2*d))+a+b)*COS(d/n) 

1160    LET l1=(a*COS(h/c*(2*d))+a+b)*COS(d/n) 

1170    LET sum=sum+SQR((st)^2+(l2-l1)^2) 

1180 NEXT h 

1190 LET y=(a*COS(h/c*(2*d))+a+b)*SIN(d/n) 

1200 PLOT POINTS:sum+100,-y+200 

 

1210 LET st=(2*a+b)/100000 

1220 FOR h=0 TO 2*a+b-st STEP st 

1230    LET y=SQR((2*a+b)^2-(h)^2)*SIN(d/n) 

1240    PLOT POINTS:sum+100,-y+200 

1250    LET l2=SQR((2*a+b)^2-(h+st)^2)*COS(d/n) 

1260    LET l1=SQR((2*a+b)^2-(h)^2)*COS(d/n) 

1270    LET sum=sum+SQR((st)^2+(l2-l1)^2) 

1280 NEXT h 

図 3-5 

sum1 sum2 sum3 

sum1=√(Δh)2  + (Δℓ)2 

Δh 

 

Δℓ 

 

ℓ 

h 
1

100000
𝑐 

2

100000
𝑐 

3

100000
𝑐 

o 

＜プログラム 3-1＞ 

3. 結果 

 このプログラムを、例えば a=50,b=50,c=300,n=8 で実行すると、図 3-6 の展開図が表示され、それを 8 枚用意し切り取って組

み立てると画像 3-1、3-2 のように花瓶ができあがります。 

 

 

 

 

 また、ステップ実行という機能(プログラムの各段階におい

て、各変数の値を知ることができる。)を使うことによって、

展開図の最底部にあたる点について、h=2×a+b(2×a+b は下

部のみでの高さなので、h の下部における最終の値となる。)

のときの sum の値と、y=0 で点が打てているということがき

ちんと表示され、想定通りの値で最底部の点が打てているこ

とが確認できました。点を打つ間隔を広くするとそのことが

よくわかるので、その図も以下に掲載します。(図 3-7) 

(st=c/100 にして、a=50,b=50,c=300,n=8 で実行した。) 

 
 

第 4 章 結論 

 今回の研究では回転体を展開可能な立体にモデ

ル化し、その展開図をコンピュータで出力するこ

とと、その立体に合わせて画像を曲げることに成

功しました。また、今回の研究からあらゆる曲面

を展開可能な立体にモデル化することや、さらに

その立体に合わせて画像を曲げて貼り付けること

ができるようになるのではないかと予想されます。 

 

第 5 章 感想 

 今回の研究のメンバーは三人とも元から数学が

好きではありましたが、今回の研究で数学やコン

ピュータに対する見方がさらに変化したように思

います。 

 プログラムを作成する中では、想定と違うもの

が出力されることがよくありました。その度に｢ど

こかが違う。一体どこが間違っているんだ。｣と私

たちは苦しみました。しかし一から論理を見直せ

ば間違いに気づくことができました。そうして今

回私たちは、数学は落ち着いて論理的に考えるこ

とが基本ながら大切なのだと思いました。また、

そうして問題が解決したときには、大きな達成感

があることを知り、数学や論理的な問題に取り組

むことをさらに楽しいと思えるようになりました。 

 数学において答えはひとつです。しかし数学に

どのように取り組むか、ということは人それぞれ

だと思います。数学は苦しい部分もありますが、

取り組めば取り組むほど新たな見方ができるよう

になります。それが、数学を”使う”ということの醍

醐味ではないかと思いました。 

 

図 3-6 

図 3-7 

↑画像 3-1   画像 3-2(底の様子)→ 

sum の正の方向 

y の正の方向 


